```
/*Evaluation of FLuorometholone as Adjunctive MEdical Therapy for Trachomatous Trichiasis Surgery: The FLAME Randomized
Controlled Clinical Trial*/
/*purpose: Comparison of primary outcome (PTT) between randomized treatment groups*/
/*Note: raw datasets in SAS library 'in' corresponds to the shared data file in xlsx format, their relationship are as below:
in.eligibility2 - EL
IN.RANDOMIZATION - RT
in.PRIMARY_END_W4 - week 4 record in PE
in.PRIMARY_END_M6 - month 6 record in PE
in.PRIMARY_END_M12 - month 12 record in PE
in.EYEEXAM_BASELINE - BE
in.SURGERY - SI
in.Adherence_w4 - TA*/

/*set up SAS library, TLF path, TLF title, format catalog which will be used in later analysis*/
%include "setup.sas";
libname in "&_root_in";
libname fm "&_root_fm";
libname data "&_root_data";
%let TLF_path = &_root_output\tb3_primary_endpoints.rtf;
%let TLF_title = Table 3:  Comparison of primary outcome between randomized treatment groups;

OPTIONS FMTSEARCH=(fm.fmsurgery fm.fmbaseexam fm.fmadherence fm.fmelig fm.fmprimary);
option mprint;

/*set up variable format for visualization*/
proc format;
        value gf  0 = "Placebo" 1 = "FML";
```

value bif **0** = "Unilateral" **1** = "Bilateral";

value compliancef **1** = "Very good" **2** = "Good" **3** = "Moderate" **4** = "Bad" **5** = "Very bad" **6**= "Not use";

value compliance_

1 = "Non-compliant: bottle weight change is known and <=75% of expected doses"

2 = "Compliant: bottle weight change indicates > 75% of expected doses and the medication diary indicates > 75% of expected doses"

3 = "Compliant: medication diary indicates >75% of expected doses and the bottle weight change is unknown"

4 = "Compliant: bottle weight change indicates >75% of expected doses and the medication diary is unknown"

5.1 = "Compliant: medication diary is >75% of expected doses"

5.2 = "Non-compliant: medication diary is known and <=75% of expected doses"

6.1 = "Compliance: medication diary is unknown and the bottle weight is unknown, Self-reported adherence (worst among two eyes) is very good"

6.2 = "Non-compliant: medication diary is unknown and the bottle weight is unknown, Self-reported adherence (worst among two eyes) is very bad - good"

6.3 = "Missing: medication diary is unknown and the bottle weight is unknown, Self-reported adherence (worst among two eyes) is missing";

value ynf **1**='Yes' **0**='No';

value ctrich_sev **0**='Non-severe' **1**='Severe**';

value cpapilla **0**='Absence' **1**='Presence';

value ccompliance **0**='<=75%' **1**='>75%';

value sisurgery_ul **0**='BLTR' **1**='PLTR';

value pval (default=**8**)

low - <**0.00095** = '<0.001'

0.00095 - <**0.0095** = [**8.3**]

0.0095 - <**0.045** = [**8.2**]

0.045 - <**0.0495** = [**8.3**]

0.0495 - <**0.04995** = [**8.4**]

```
                0.04995 - <0.05 = '~<0.05'
                0.05 = '0.05'
                0.05< - <0.05005 = '~>0.05'
                0.05005 - <0.0505 = [8.4]
                0.0505 - <0.055 = [8.3]
                0.055 - 0.99 = [8.2]
                0.99< - high = [8.2];
run;

* combine covariate used for analysis;
data PRIMARY_END_W4;
set in.PRIMARY_END_W4;
run;

data PRIMARY_END_M6;
set in.PRIMARY_END_M6;
run;

data PRIMARY_END_M12;
set in.PRIMARY_END_M12;
run;

data EYEEXAM_BASELINE;
set in.EYEEXAM_BASELINE;
run;

data SURGERY;
```

```
set in.SURGERY;
run;

options mprint;
/*reshape the dataset to be eye level*/
%proc_reshape(indata=WORK.PRIMARY_END_W4,memname="PRIMARY_END_W4",outdata=PRIMARY_END_W4_);
%proc_reshape(indata=WORK.PRIMARY_END_M6,memname="PRIMARY_END_M6",outdata=PRIMARY_END_M6_);
%proc_reshape(indata=WORK.PRIMARY_END_M12,memname="PRIMARY_END_M12",outdata=PRIMARY_END_M12_);
%proc_reshape(indata=WORK.EYEEXAM_BASELINE,memname="EYEEXAM_BASELINE",outdata=EYEEXAM_BASELINE_);
%proc_reshape(indata=WORK.SURGERY,memname="SURGERY",outdata=SURGERY_);

data PRIMARY_END_W4_1;
        set PRIMARY_END_W4_;
        if peeye_2 =1 or peeye_3 =1 or peeye_4 =1 then have=1;
        else if peeye_0=1 then have=0;
        else have=.;
run;

data PRIMARY_END_M6_1;
        set PRIMARY_END_M6_;
        if peeye_2 =1 or peeye_3 =1 or peeye_4 =1 then have=1;
        else if peeye_0=1 then have=0;
        else have=.;
run;

data PRIMARY_END_M12_1;
        set PRIMARY_END_M12_;
```

```
        if peeye_2 =1 or peeye_3 =1 or peeye_4 =1 then have=1;
        else if peeye_0=1 then have=0;
        else have=.;
run;

data fas;
        set data.fas;
        if reye_eligibel=1 then do; eye='OD'; output; end;
        if leye_eligibel=1 then do; eye='OS'; output; end;
run;

proc sql;
        create table Primary_endpoint as
        select r.subjid, r.eye, r.treat,
        case when c.subjid^='' then 1 else 0 end as w4fl, c.have as have_w4,c.peeye_2 as peeye_2_w4, c.peeye_3 as
peeye_3_w4, c.peeye_4 as peeye_4_w4,
        case when b.subjid^='' then 1 else 0 end as m6fl, b.have as have_M6,b.peeye_2 as peeye_2_M6, b.peeye_3 as
peeye_3_M6, b.peeye_4 as peeye_4_M6,
        case when a.subjid^='' then 1 else 0 end as m12fl, a.have as have_M12,a.peeye_2 as peeye_2_M12, a.peeye_3 as
peeye_3_M12, a.peeye_4 as peeye_4_M12,
        case when c.subjid^='' or b.subjid^='' or a.subjid^='' then 1 else 0 end as fl,
        case when c.have=1 or b.have=1 or a.have=1 then 1
                else 0 end as have,
        case when c.subjid^='' or b.subjid^='' then 1 else 0 end as earlyfl,
        case when c.have=1 or b.have=1 then 1
                else 0 end as have_early,
        case when a.subjid^='' then 1 else 0 end as latefl,
```

```
        case when a.have=1 and b.have^=1 and c.have^=1 then 1
                else 0 end as have_late,
        /*to meet requirement of HIPAA, name (sisiname2) and certification (sisicert2) of surgeon is removed from the public shared dataset*/
        sisicert2,sisiname2,
        /*for surgeon that only have a small number of surgery (<50), cluster them into 1 group*/
        case when sisicert2 in (/*surgeon license removed*/) then 'Other'
                else sisicert2 end as sisicert3
        from fas as r
        left join WORK.PRIMARY_END_M12_1 as a on r.subjid = a.subjid and r.eye = a.eye
        left join WORK.PRIMARY_END_M6_1 as b on r.subjid = b.subjid and r.eye = b.eye
        left join WORK.PRIMARY_END_w4_1 as c on r.subjid = c.subjid and r.eye = c.eye
        left join in.surgery as d on r.subjid=d.subjid;
quit;

/*Build subgroup of analysis*/
/* ============ Prepare strata variable 1 - treatment adherence rate ================== */
/*
```

Classify adherence using a binary variable of roughly >75% of expected doses given.

- If the medication diary is all 0's, consider the value as unknown. These are likely people who did not fill out the diary correctly.

Apply these rules in a hierarchy:

1. If the bottle weight change is known and =75% of expected doses, then the patient is non-compliant.
2. If the bottle weight change indicates > 75% and the medication diary indicates > 75% then compliant.
3. If the medication diary indicates >75% and the bottle weight change is unknown, then compliant.
4. If the bottle weight change indicates >75% and the medication diary is unknown, then compliant.
5. The adherence by bottle weight is higher than adherence by medication diary, particularly in unilateral surgery.

This may be due to treating both eyes in unilateral surgery as well as missing the eye when applying the drop and needing another drop for both unilateral and bilateral surgery.
Use the medication diary if it is known. If the medication diary is =75%, then non-compliant.
6. If the medication diary is unknown and the bottle weight is unknown, use the self-report and consider only Very Good as compliant.
*/

```
* 1) Adherence date eye level;
proc sql;
create table Adh_study_r as
        select a.subjid, "OD" as eye,
        taeyedrops_r as taeyedrops,tadrops_compliance_r as tadrops_compliance
        from in.Adherence_w4 as a
        left join in.eligibility2 as c on a.subjid = c.subjectid
        where c.elelig_r = 1;

create table Adh_study_l as
        select a.subjid, "OS" as eye,
        taeyedrops_l as taeyedrops,tadrops_compliance_l as tadrops_compliance
        from in.Adherence_w4 as a
        left join in.eligibility2 as c on a.subjid = c.subjectid
        where c.elelig_l = 1;
quit;

data Adh_studyeye;
        set Adh_study_r Adh_study_l;
        if  tadrops_compliance = . then tadrops_compliance = 6;
run;
```

```
*2) Surgery Data of study eye;
proc sql;
        create table surgery_OS as
        select subjid,sisurgerydtc_lul as surgery_dt, "OS" as eye
        from  IN.SURGERY as s
        left join  IN.ELIGIBILITY2 as e on e.subjectid = s.subjid
        where  elelig_l = 1;

        create table surgery_OD as
        select subjid,sisurgerydtc_rul as surgery_dt, "OD" as eye
        from  IN.SURGERY as s
        left join  IN.ELIGIBILITY2 as e on e.subjectid = s.subjid
        where  elelig_r = 1;
quit;
data surgery_dt0 ;
        set surgery_OS surgery_OD;
  if surgery_dt=. then surdtfl=0;
  else surdtfl=1;
run;

* select the earlist surgery date;
proc sort data = surgery_dt0 out = surgery_dt (drop = eye);
        by subjid descending surdtfl surgery_dt;
run;
data surgery_dt(drop=surdtfl);
        set surgery_dt;
```

```
        by subjid;
        if first.subjid;
run;

proc sql;
create table Adherence_person as
        select a.*,
        case when e.sisurgery_rul in (0,1) and e.sisurgery_lul in (0,1) then 1
                else 0 end as bi format = bif.,
        case when e.sisurgery_rul in (0,1) and e.sisurgery_lul in (0,1) then tatimes_used_morning_l+
tatimes_used_morning_r+tatimes_used_night_r+tatimes_used_night_l
                when e.sisurgery_rul in (0,1) and e.sisurgery_lul not in (0,1) then tatimes_used_morning_r+tatimes_used_night_r
                when e.sisurgery_rul not in (0,1) and e.sisurgery_lul in (0,1) then tatimes_used_morning_l+tatimes_used_night_l
end as total_times,    surgery_dt,
        talastday-surgery_dt+1 as Duration,
        case when tavisitdtc-surgery_dt>28+4 or .<tavisitdtc-surgery_dt<28-4 then 28
                 else tavisitdtc-surgery_dt end as expected_days,
        rtmedweight as Baseline_weight,
        (tareturn_weight - rtmedweight) as change_weight
        from in.Adherence_w4 as a
        left join in.eligibility2 as c on a.subjid = c.subjectid
        left join  IN.RANDOMIZATION as d on a.subjid= d.subjid
        left join  IN.SURGERY as e on a.subjid= e.subjid
        left join surgery_dt as dt on a.subjid = dt.subjid;
quit;

data Adherence_person;
```

```
	set Adherence_person;
	if bi = 1 then do;
		ad_scale1 = -change_weight *100 / (expected_days * 4 *0.05);
		ad_scale2 = total_times *100 / (expected_days * 4 );
		expected_weightchange = - expected_days * 4 *0.05;
	end;
	else if bi = 0 then do;
		ad_scale1 = -change_weight *100 / (expected_days * 2 *0.05);
		ad_scale2 = total_times *100 / (expected_days * 2 );
		expected_weightchange = - expected_days * 2 *0.05;
	end;
run;

data Adherence_person;
	set Adherence_person;
	if ad_scale1 < 0 then ad_scale1 = . ;
	if ad_scale2 < 0 then ad_scale2 = . ;
run;

proc sql;
	create table SR_Adh as
	select distinct subjid,  max(tadrops_compliance) as tadrops_compliance format =compliancef.
	from WORK.ADH_STUDYEYE
	group by subjid;
quit;

proc sql;
```

```
        create table ADHERENCE_PERSON as
        select a.* , b.tadrops_compliance
        from ADHERENCE_PERSON as a
        left join SR_Adh as b
        on a.subjid = b.subjid;
quit;

proc sql;
create table final0 as
select a.*,
        case when c.sisurgery_rul ne 9 and c.sisurgery_lul ne 9 then 'B'
                when c.sisurgery_rul ne 9 then 'R'
                when c.sisurgery_lul ne 9 then 'L' end as eye1
from adherence_person as a
left join in.surgery as c
on a.subjid=c.subjid
order by subjid;
quit;
data final;
format compliance compliance_.;
set final0;
if ad_scale2=0 and eye1='B' and (taeyedrops_l=1 or taeyedrops_r=1) then ad_scale2=.;
IF ad_scale2=0 and eye1='L' and taeyedrops_l=1 then ad_scale2=.;
if ad_scale2=0 and eye1='R' and taeyedrops_r=1 then ad_scale2=.;
/*If the medication diary is all 0's, consider the value as unknown
need to double check the taeyedrops_l or taeyedrops_r, it is also possible that the patient does not take the drop
*/
```

```
if .<ad_scale1<=75 then compliance=1;
else if ad_scale1>75 and ad_scale2>75 then compliance=2;
else if ad_scale1=. and ad_scale2>75 then compliance=3;
else if ad_scale1>75 and ad_scale2=. then compliance=4;
else if ad_scale2>75 then compliance=5.1;
else if .<ad_scale2<=75 then compliance=5.2;
else if ad_scale2=. and ad_scale2=. then do;
        if tadrops_compliance=1 then compliance=6.1;
        else if tadrops_compliance^=6 then compliance=6.2;
        else if tadrops_compliance=6 then compliance=6.3;
end;
run;
proc freq data=final;
tables compliance/ missing nopercent nocol;
label compliance='Treatment compliance: criteria';
run;

/*combine the strata variable for subgroup analysis*/
proc sql;
        create table data.tbl3_data as
        select a.*,
                case when sum(becornea_ul,bemedial_ul,belateral_ul)>=6 or beepilation_ul in (2,3) then 1
                        when .<sum(becornea_ul,bemedial_ul,belateral_ul)<6 and beepilation_ul in (0,1) then 0 end as
ctrich_sev,
                /*betrachint as cconjinflam format=8.,*/
                case when bepapilla_ul=0 then 0
                        when bepapilla_ul in (1,2,3) then 1 end as cpapilla,
```

```
                case when compliance in (1,5.2,6.2) then 0
                        when compliance in (2,3,4,5.1,6.1) then 1 end as ccompliance,
                case when e.sisurgery_ul in (0,1) then e.sisurgery_ul end as sisurgery_ul format=8.
        from Primary_endpoint as a
        left join EYEEXAM_BASELINE_ as c
        on a.subjid=c.subjid and a.eye=c.eye
        left join final as d
        on a.subjid=d.subjid
        left join surgery_ as e
        on a.subjid=e.subjid and a.eye=e.eye;
quit;

/*calculate the N and primary endpoint number (percentage)*/
%macro macro_freq_perc(input=,have=);
proc freq data=&input;
        tables treat*&have / out=tab_&have._1(keep= treat &have count pct_row) outpct;
        tables treat / out=tab_&have._2(keep=treat count);
run;
proc sql;
        create table tab_&have. as
        select a.treat, a.count as denominator, ifn(b.count^=.,b.count,0) as n, strip(put(calculated n,best.))||'
('||strip(put(pct_row,10.2))||')' as freq_perc
        from tab_&have._2 as a
        left join tab_&have._1(where=(&have=1)) as b
        on a.treat=b.treat;
quit;
proc datasets;
```

```
	delete tab_&have._2 tab_&have._1;
quit;
%mend;

%macro_freq_perc(input=data.tbl3_data(where=(fl=1)),have=have);
%macro_freq_perc(input=data.tbl3_data(where=(w4fl=1)),have=have_w4);
%macro_freq_perc(input=data.tbl3_data(where=(m6fl=1)),have=have_m6);
%macro_freq_perc(input=data.tbl3_data(where=(m12fl=1)),have=have_m12);
%macro_freq_perc(input=data.tbl3_data(where=(earlyfl=1)),have=have_early);
%macro_freq_perc(input=data.tbl3_data(where=(latefl=1)),have=have_late);

%macro macro_freq_perc_sub(input=,var=);
proc freq data=&input;
	tables &var.*treat*have / out=tab_&var._1(keep=&var. treat have count pct_row) outpct;
	tables &var.*treat / out=tab_&var._2(keep=&var. treat count) outpct;
run;
proc sql;
	create table tab_&var. as
	select a.&var., a.treat, a.count as denominator, ifn(b.count^=.,b.count,0) as n, round((calculated
n/a.count),0.0001)*100 as perc, strip(put(calculated n,best.))||' ('||strip(put(calculated perc,best.))||')' as freq_perc
	from tab_&var._2 as a
	left join tab_&var._1(where=(have=1)) as b
	on a.&var.=b.&var. and a.treat=b.treat
	where a.&var.^=.;
quit;
proc datasets;
	delete tab_&var._2 tab_&var._1;
```

```
quit;
%mend;

%macro_freq_perc_sub(input=data.tbl3_data(where=(fl=1)),var=ctrich_sev);
%macro_freq_perc_sub(input=data.tbl3_data(where=(fl=1)),var=cpapilla);
%macro_freq_perc_sub(input=data.tbl3_data(where=(fl=1)),var=ccompliance);
%macro_freq_perc_sub(input=data.tbl3_data(where=(fl=1)),var=sisurgery_ul);

%macro macro_statistics(input=,have=,surgeon_cluster=0,surg_bar=0);
/*when there is no enough record per surgeon category and outcome variable category, there will be model converging issue
cluster some of the surgeon category to solve the issue*/
%if &surgeon_cluster=1 %then %do;
proc freq data=&input;
	tables sisicert3*&have / out=surgeon;
run;
proc sql;
	create table surgeon_ as
	select distinct sisicert3,
				min(count) as fl1,
				count(distinct &have) as fl2
	from surgeon(where=(sisicert3^='' and &have is not missing))
	group by sisicert3;
quit;

proc sql;
	create table input as
	select a.*,
```

```
                case when a.sisicert2^='' and (fl1<=&surg_bar or fl2<2) then 'Other'
                        else a.sisicert3 end as sisicert4
        from &input as a
        left join surgeon_ as b
        on a.sisicert3=b.sisicert3;
quit;
proc sql;
        select distinct sisicert3
        from input
        where sisicert3^=sisicert4;
quit;
%end;
%if &surgeon_cluster=0 %then %do;
data input;
        length sisicert4 $200;
        set &input;
        sisicert4=sisicert3;
run;
%end;
/*calculate p-value and odds ratio for primary analysis*/
proc genmod data=input descending;
        class subjid treat(ref='Placebo') sisicert4;
        model &have = treat sisicert4 / dist=bin type3;
        repeated sub=subjid/type=ind covb corrw corrb;
        ods output Type3  = type3(where=(upcase(source)='TREAT')) diffs=diffs ConvergenceStatus=ConvergenceStatus;
        lsmeans treat/ e exp diff cL ilink;
run;
```

```
 /*calculate risk difference for primary analysis*/
%margins(data=input,
     class=subjid treat sisicert4,
          response=&have,
          roptions= event='1',
          model= treat sisicert4,
          dist=binomial,
          geesubject=subjid,
          geecorr=IND,
     margins=treat,
     balanced=sisicert4,
          diff=ALL,
     options=cl);

data diffs;
          length OR $50;
          set diffs;
          OR=strip(put(round(ExpEstimate,0.01),10.2))||' ('||strip(put(round(LowerExp,0.01),10.2))||',
'||strip(put(round(UpperExp,0.01),10.2))||')';
run;

data diffs_rd;
          length RD $50;
          set _diffspm;
          RD=strip(put(round(Estimate,0.01),10.2))||' ('||strip(put(round(Lower,0.01),10.2))||',
'||strip(put(round(Upper,0.01),10.2))||')';
run;
```

```
data ConvergenceStatus;
	length reason_ $200;
	set ConvergenceStatus(where=(Status^=0)) end=eof;
	if _n_=1 then reason_=strip(reason);
	else if _n_>1 then reason_=strip(reason_)||';'||strip(reason);
	if eof then output;
run;

data ConvergenceStatus_rd;
	length reason_ $200;
	set ConvergenceStatus_rd(where=(Status^=0)) end=eof;
	if _n_=1 then reason_=strip(reason);
	else if _n_>1 then reason_=strip(reason_)||';'||strip(reason);
	if eof then output;
run;

proc sql;
	create table temp_&have. as
	select case when d.reason_='' then OR
				else 'NA' end as OR,
		case when e.reason_='' then RD
				else 'NA' end as RD,
		case when d.reason_='' then put(ProbChiSq,pval.)
				else 'NA' end as pvalue
	from diffs_rd as a
	full join diffs as b on 1
```

```
        full join type3 as c on 1
        full join ConvergenceStatus as d on 1
        full join ConvergenceStatus_rd as e on 1;
quit;

proc datasets;
        delete type3 diffs diffs_rd input surgeon ConvergenceStatus ConvergenceStatus_rd _diffspm _margins;
quit;
%mend;

%macro_statistics(input=data.tbl3_data(where=(fl=1)),have=have);
%macro_statistics(input=data.tbl3_data(where=(w4fl=1)),have=have_w4,surgeon_cluster=1,surg_bar=0);
%macro_statistics(input=data.tbl3_data(where=(m6fl=1)),have=have_m6,surgeon_cluster=1,surg_bar=0);
%macro_statistics(input=data.tbl3_data(where=(m12fl=1)),have=have_m12);
%macro_statistics(input=data.tbl3_data(where=(earlyfl=1)),have=have_early);
%macro_statistics(input=data.tbl3_data(where=(latefl=1)),have=have_late,surgeon_cluster=1,surg_bar=0);

%macro macro_statistics_sub(input=,var=,surgeon_cluster=0,surg_bar=0);
/*when there is no enough record per surgeon category and outcome variable category, there will be model converging issue
cluster some of the surgeon category to solve the issue*/
%if &surgeon_cluster=1 %then %do;
proc freq data=&input;
        tables &var*have*sisicert3 / out=surgeon;
run;
proc sql;
        create table surgeon_ as
        select distinct sisicert3,
```

```
                                        min(count) as fl1,
                                        count(distinct put(have,best.)||'-'||put(&var,best.)) as fl2
        from surgeon(where=(sisicert3^='' and have is not missing and &var is not missing))
        group by sisicert3;
quit;

proc sql;
        create table input as
        select a.*,
                case when a.sisicert2^='' and (fl1<=&surg_bar or fl2<4) then 'Other'
                        else a.sisicert3 end as sisicert4
        from &input as a
        left join surgeon_ as b
        on a.sisicert3=b.sisicert3;
quit;
proc sql;
        select distinct sisicert3
        from input
        where sisicert3^=sisicert4;
quit;
%end;
%if &surgeon_cluster=0 %then %do;
data input;
        length sisicert4 $200;
        set &input;
        sisicert4=sisicert3;
run;
```

```
%end;
/*calculate p-value and odds ratio for subgroup analysis*/
proc genmod data=input descending;
        class subjid treat(ref='Placebo') &var(ref='0') sisicert4;
        model have = treat &var treat*&var. sisicert4/ dist=bin type3;
        repeated sub=subjid/type=ind covb corrw;
        ods output Type3  = type3 diffs=diffs ConvergenceStatus=ConvergenceStatus;
        lsmeans treat*&var./ e exp diff cL ilink;
        estimate 'treat FML - subgroup 1' intercept 1 treat 1 0 &var. 1 0 treat*&var. 1 0 0 0/ e;
        estimate 'treat FML - subgroup 0' intercept 1 treat 1 0 &var. 0 1 treat*&var. 0 1 0 0/ e;
        estimate 'treat Placebo - subgroup 1' intercept 1 treat 0 1 &var. 1 0 treat*&var. 0 0 1 0/ e;
        estimate 'treat Placebo - subgroup 0' intercept 1 treat 0 1 &var. 0 1 treat*&var. 0 0 0 1/ e;
run;
/*calculate risk difference for subgroup analysis*/
%margins(data=input,
    class=subjid treat &var sisicert4,
            response=have,
            roptions=event='1',
            model=treat &var treat*&var. sisicert4,
            dist=binomial,
    geesubject=subjid,
            geecorr=IND,
    margins=treat,
    at=&var,
    balanced=sisicert4,
    diff=all,
    alpha=0.05,options=cl);
```

```
data diffs;
	length OR $50;
	set diffs;
	if &var.=_&var.;
	OR=strip(put(round(ExpEstimate,0.01),10.2))||' ('||strip(put(round(LowerExp,0.01),10.2))||',
'||strip(put(round(UpperExp,0.01),10.2))||')';
run;

data diffs_rd;
	length RD $50;
	set _diffspm;
	RD=strip(put(round(Estimate,0.01),10.2))||' ('||strip(put(round(Lower,0.01),10.2))||',
'||strip(put(round(Upper,0.01),10.2))||')';
run;

data ConvergenceStatus;
	length reason_ $200;
	set ConvergenceStatus(where=(Status^=0)) end=eof;
	if _n_=1 then reason_=strip(reason);
	else if _n_>1 then reason_=strip(reason_)||';'||strip(reason);
	if eof then output;
run;

data ConvergenceStatus_rd;
	length reason_ $200;
	set ConvergenceStatus_rd(where=(Status^=0)) end=eof;
```

```
	if _n_=1 then reason_=strip(reason);
	else if _n_>1 then reason_=strip(reason_)||';'||strip(reason);
	if eof then output;
run;

proc sql;
	create table temp_&var. as
	select a.&var,
		case when d.reason_='' then OR
			else 'NA' end as OR,
		case when e.reason_='' then RD
			else 'NA' end as RD,
		case when d.reason_='' then put(ProbChiSq,pval.)
			else 'NA' end as pvalue
	from diffs_rd as a
	left join diffs as b
	on a.&var.=b.&var.
	left join type3(where=(index(Source,'*'))) as c
	on 1=1
	left join ConvergenceStatus as d
	on 1
	left join ConvergenceStatus_rd as e
	on 1;
quit;
proc datasets;
	delete type3 diffs diffs_rd ConvergenceStatus ConvergenceStatus_rd _diffspm _margins;
quit;
```

```
%mend;

%macro_statistics_sub(input=data.tbl3_data(where=(fl=1)),var=ctrich_sev);
%macro_statistics_sub(input=data.tbl3_data(where=(fl=1)),var=cpapilla);
%macro_statistics_sub(input=data.tbl3_data(where=(fl=1)),var=ccompliance);
%macro_statistics_sub(input=data.tbl3_data(where=(fl=1)),var=sisurgery_ul);

/*combine all output for primary analysis*/
%macro combine_tab(var=,tab_order=,questionlabel=);
proc sql;
	create table final_&var as
	select &tab_order as tab_order, "&questionlabel" as questionlabel,
	strip(put(a.denominator,best.)) as N_placebo, a.freq_perc as E_placebo, strip(put(b.denominator,best.)) as N_fml,
b.freq_perc as E_fml,
	c.OR, c.RD, strip(c.pvalue) as pval
	from tab_&var(where=(treat=0)) as a,tab_&var(where=(treat=1)) as b,temp_&var as c;
quit;
%mend;
%combine_tab(var=have,tab_order=1,questionlabel=PTT at any time during one-year follow-up);
%combine_tab(var=have_w4,tab_order=2,questionlabel=  PTT present at 4 weeks);
%combine_tab(var=have_m6,tab_order=3,questionlabel=  PTT present at 6 months);
%combine_tab(var=have_m12,tab_order=4,questionlabel=  PTT present at 12 months);
%combine_tab(var=have_early,tab_order=5,questionlabel=  Early-onset PTT (incident within the 6-month visit) );
%combine_tab(var=have_late,tab_order=6,questionlabel=  Late-onset PTT (incident after the 6-month visit));

/*combine all output for subgroup analysis*/
%macro combine_tab_sub(var=,tab_order=,questionlabel=);
```

```
proc sql;
	create table final_&var as
	select &tab_order as tab_order, ' '||strip(put(a.&var,&var..)) as questionlabel, 1-a.&var as answerorder,
	strip(put(a.denominator,best.)) as N_placebo, a.freq_perc as E_placebo, strip(put(b.denominator,best.)) as N_fml,
b.freq_perc as E_fml,
	c.OR, c.RD, case when a.&var=1 then 'Interaction P='||strip(c.pvalue) else '' end as pval
	from tab_&var(where=(treat=0)) as a
	full join tab_&var(where=(treat=1)) as b
	on a.&var=b.&var
	full join temp_&var as c
	on a.&var=c.&var
	order by a.&var;
quit;
data header;
	length questionlabel $200;
	answerorder=-99;
	questionlabel="&questionlabel";
	tab_order=&tab_order;
run;
data final_&var;
	length questionlabel N_placebo N_fml E_placebo E_fml OR RD pval $200;
	set header final_&var;
run;
proc datasets;
	delete header;
quit;
%mend;
```

```
%combine_tab_sub(var=ctrich_sev,tab_order=7,questionlabel=Baseline upper eyelid trichiasis severity);
%combine_tab_sub(var=cpapilla,tab_order=8,questionlabel=Baseline conjunctival inflammation);
%combine_tab_sub(var=ccompliance,tab_order=9,questionlabel=%str(Treatment adherence~{super #}));
%combine_tab_sub(var=sisurgery_ul,tab_order=10,questionlabel=%str(Surgery Type));

data header;
        length questionlabel $200;
        tab_order=6.5;
        questionlabel="Subgroup Analysis for the outcome of PTT anytime during one year follow-up";
run;
data data.tbl3_result;
length questionlabel N_placebo N_fml E_placebo E_fml OR RD pval $200.;
set final_: header;
if tab_order in (1,2,3,4) then section_order=1;
else if tab_order in (5,6) then section_order=2;
else if tab_order>6 then section_order=3;
run;
proc sort data=data.tbl3_result; by section_order tab_order answerorder; run;

/*calculate header and footnote #*/
proc sql;
select count(*) into :placebo_eye
from data.tbl3_data
where treat=0 and fl=1;
select count(*) into :fml_eye
from data.tbl3_data
```

```
where treat=1 and fl=1;
quit;
%put &placebo_eye;
%put &fml_eye;

proc sql;
select count(*) into :missing_0
from data.tbl3_data
where treat=0 and fl=1 and ccompliance=.;
select count(*) into :missing_1
from data.tbl3_data
where treat=1 and fl=1 and ccompliance=.;
quit;
%put &missing_0;
%put &missing_1;

/*table visualization*/
options nodate nonumber;
options orientation=landscape;
ods rtf file = "&TLF_path";
ods escapechar='~';
ods rtf text="~S={just=l font_size=10pt font_weight= bold} &TLF_title";
proc report data=data.tbl3_result nowd spanrows split='|' missing style(column)={background=white fontsize=9pt}

style(header)={background=white fontsize=9pt fontweight=medium};
        columns section_order tab_order questionlabel answerorder ("Placebo|(n=%trim(&placebo_eye) study eyes)"
N_placebo E_placebo)  ("FLuorometholone|(n=%trim(&fml_eye) study eyes)" N_fml E_fml)  ("" RD OR pval);
```

```
        define section_order  / ' ' order order=internal noprint;
        define tab_order  / ' ' order order=internal noprint;
        define questionlabel  / ' ' left style(column)={cellwidth=1.8in asis=ON};
        define answerorder / ' ' order order=internal noprint;
        define N_: / '# of eyes' center style(column)={cellwidth=0.75in};
        define E_: / 'PTT (%)' center style(column)={cellwidth=1.2in};
        define RD /'Risk Difference|(95% CI)*' center style(column)={cellwidth=1.5in};
        define OR /'Odds Ratio|(95% CI)*' center style(column)={cellwidth=1.5in};
        define pval /'p-value*' center style(column)={cellwidth=0.7in};
        compute after section_order ;
                line '' ;
         endcomp ;
run ;
ods rtf text="~S={just=l font_size=9pt} PTT=Postoperative trachomatous trichiasis.";
ods rtf text="~S={just=l font_size=9pt} *From generalized regression models adjusting for surgeon that account for inter-eye
correlation.";
ods rtf text="~S={just=l font_size=9pt} ** Severe was defined as total number of lashes =6, or epilation =1/3.";
ods rtf text="~S={just=l font_size=9pt} #%trim(&missing_0) eyes in the placebo group and %trim(&missing_1) eyes in the
FLuorometholone group were excluded from the analysis due to missing data on treatment adherence.";
ods rtf close;
```